

NAG Fortran Library Routine Document

F07MJF (SSYTRI/DSYTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07MJF (SSYTRI/DSYTRI) computes the inverse of a real symmetric indefinite matrix A , where A has been factorized by F07MDF (SSYTRF/DSYTRF).

2 Specification

```
SUBROUTINE F07MJF(UPLO, N, A, LDA, IPIV, WORK, INFO)
ENTRY          ssytri (UPLO, N, A, LDA, IPIV, WORK, INFO)
INTEGER        N, LDA, IPIV(*), INFO
real          A(LDA,*), WORK(*)
CHARACTER*1     UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To compute the inverse of a real symmetric indefinite matrix A , this routine must be preceded by a call to F07MDF (SSYTRF/DSYTRF), which computes the Bunch–Kaufman factorization of A .

If UPLO = 'U', $A = PUDU^T P^T$ and A^{-1} is computed by solving $U^T P^T XPU = D^{-1}$ for X .

If UPLO = 'L', $A = PLDL^T P^T$ and A^{-1} is computed by solving $L^T P^T XPL = D^{-1}$ for X .

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

- 1: UPLO – CHARACTER*1 *Input*
On entry: indicates how A has been factorized as follows:
 if UPLO = 'U', $A = PUDU^T P^T$, where U is upper triangular;
 if UPLO = 'L', $A = PLDL^T P^T$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: A(LDA,*) – ***real*** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: details of the factorization of A , as returned by F07MDF (SSYTRF/DSYTRF).

On exit: the factorization is overwritten by the n by n symmetric matrix A^{-1} . If UPLO='U', the upper triangle of A^{-1} is stored in the upper triangular part of the array; if UPLO = 'L', the lower triangle of A^{-1} is stored in the lower triangular part of the array.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07MJF (SSYTRI/DSYTRI) is called.

Constraint: $LDA \geq \max(1, N)$.

5: IPIV(*) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, N)$.

On entry: details of the interchanges and the block structure of D , as returned by F07MDF (SSYTFR/DSYTFR).

6: WORK(*) – *real* array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, N)$.

7: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , d_{ii} is exactly zero; D is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies a bound of the form

$$|DU^T P^T XPU - I| \leq c(n)\epsilon(|D||U^T|P^T|X|P|U| + |D||D^{-1}|) \text{ if UPLO = 'U', or}$$

$$|DL^T P^T XPL - I| \leq c(n)\epsilon(|D||L^T|P^T|X|P|L| + |D||D^{-1}|) \text{ if UPLO = 'L',}$$

where $c(n)$ is a modest linear function of n , and ϵ is the *machine precision*.

8 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^3$.

The complex analogues of this routine are F07MWF (CHETRI/ZHETRI) for Hermitian matrices and F07NWF (CSYTRI/ZSYTRI) for symmetric matrices.

9 Example

To compute the inverse of the matrix A , where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

Here A is symmetric indefinite and must first be factorized by F07MDF (SSYTRF/DSYTRF).

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07MJF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER      NIN, NOUT
PARAMETER    (NIN=5,NOUT=6)
INTEGER      NMAX, LDA, LWORK
PARAMETER    (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*      .. Local Scalars ..
INTEGER      I, IFAIL, INFO, J, N
CHARACTER    UPLO
*      .. Local Arrays ..
real        A(LDA,NMAX), WORK(LWORK)
INTEGER      IPIV(NMAX)
*      .. External Subroutines ..
EXTERNAL      ssytrf, ssytri, X04CAF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07MJF Example Program Results'
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF
*
*      Factorize A
*
      CALL ssytrf(UPLO,N,A,LDA,IPIV,WORK,LWORK,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Compute inverse of A
*
        CALL ssytri(UPLO,N,A,LDA,IPIV,WORK,INFO)
*
*      Print inverse
*
        IFAIL = 0
        CALL X04CAF(UPLO,'Nonunit',N,N,A,LDA,'Inverse',IFAIL)
      ELSE
        WRITE (NOUT,*) 'The factor D is singular'
      END IF
    END IF
    STOP
*
  END
```

9.2 Program Data

F07MJF Example Program Data

```
4                               :Value of N
'L'                             :Value of UPLO
2.07
3.87 -0.21
4.20  1.87  1.15
-1.15  0.63  2.06 -1.81      :End of matrix A
```

9.3 Program Results

F07MJF Example Program Results

Inverse

	1	2	3	4
1	0.7485			
2	0.5221	-0.1605		
3	-1.0058	-0.3131	1.3501	
4	-1.4386	-0.7440	2.0667	2.4547
